
1 チュートリアルへようこそ!

1 チュートリアルへようこそ!
このチュートリアルでは、Qlik Sense で高度なスクリプトを作成する方法をご紹介します。

スクリプト作成の基本を習得すると、Qlik Sense にデータをロードする際に、より高度な操作を実

行できるようになります。これにはたとえば、クロス集計を使用したデータの変換、データのクレンジ

ング、QVD ファイルとして知られるQlik データファイルからのデータの作成およびロードなどが含ま

れます。

1.1 学習内容
After completing this tutorial, you should be comfortable with loading data using some of the more
advanced scripting functions in Qlik Sense.

1.2 対象ユーザー
Qlik Sense のスクリプト作成の基本を習得している必要があります。具体的には、スクリプトを使用し、データを

ロードして操作したことがある方が対象です。

未経験の方は、初心者向けチュートリアルを完了することを推奨します。

データロードエディターへのアクセス権があり、Qlik Sense Enterprise on Windows でデータをロードできる必

要があります。

上記の指示は通常、Qlik Sense Cloud Business にも適用されます。

1.3 パッケージ内容
ダウンロードしたzip パッケージには、チュートリアルを完了するために必要な次のデータファイルが含まれていま

す:

l Cutlery.xlsx
l Data.xlsx
l Events.txt
l Employees.xlsx
l Intervals.txt
l Product.xlsx
l Salesman.xlsx
l Transactions.csv
l Winedistricts.txt

また、パッケージには高度なスクリプトチュートリアルアプリも含まれています。アプリ内の追加のスクリプトセクショ

ンに、このチュートリアルで作成するその他のアプリのスクリプトが含まれています。ハブにアプリをアップロードできま

す。

1 チュートリアルへようこそ!

学習効果を最大限に高めるため、チュートリアルの説明どおりに自分自身でアプリを構築することを推奨しま

す。また、データロードを機能させるには、チュートリアルの説明どおりに、データファイルをアップロードしてデータ

ファイルに接続する必要があります。

ただし、問題が発生した場合は、アプリがトラブルシューティングに役立つ場合があります。各レッスンにどのスクリ

プトセグメントが関連しているのかを示しました。

1.4 このチュートリアルでのレッスン
Qlik Sense の使用経験により異なりますが、このチュートリアルの所要時間は3 ～ 4 時間です。トピックは、順

番に完了するよう構成されています。ただし、トピックをとばして、後で戻ることもできます。幸い、テストはありませ

ん。

データの変換

Crosstable プレフィックスの使用

Join とKeep を使用したテーブルの結合

レコード間関数の使用 : Peek、Previous、Exist
間隔の一致と反復ロード

データクレンジング

階層データの処理

QVD ファイル

1.5 追加の資料とリソース
l ≤ Qlik では、さらなる詳細情報を提供する、広範なリソースをご用意しています。

l Qlik オンラインヘルプを使用できます。

l 無料のオンラインコースを含むトレーニングは、≤ Qlik Continuous Classroom で利用できます。

l ディスカッションフォーラム、ブログなどは、≤ Qlik Community にあります。

2 LOAD とSELECT ステートメント

2 LOAD とSELECT ステートメント
LOAD およびSELECT ステートメントを使用すると、Qlik Sense にデータをロードできます。これらの各ステートメ

ントにより、内部テーブルが生成されます。LOAD はファイルからのデータのロードに使用されるのに対し、

SELECT はデータベースからのデータのロードに使用されます。

このチュートリアルでは、ファイルのデータを使用するため、LOAD ステートメントを使用します。

先行するLOAD を使用して、ロードするデータのコンテンツを操作することもできます。たとえば、LOAD ステート

メントで項目名を変更する必要があり、SELECT ステートメントでの項目名の変更を許可しないようにします。

以下の規則はデータをQlik Sense にロードする際に適用されます。

l Qlik Sense では、LOAD ステートメントで生成されるテーブルとSELECT ステートメントで生成される

テーブルに違いはありません。つまり、複数のテーブルをロードする場合、LOAD とSELECT のいずれか一

方のステートメントを使用するか、これら2 つを組み合わせて使用するかは問題にはなりません。

l Qlik Sense ロジックでは、ステートメント内またはデータベース内の元のテーブルの項目の順序は重要で

はありません。

l 項目名は大文字小文字が区別され、データテーブル間の関連付けに使用されます。このため、適切な

データモデルを作成するには、ロードスクリプトで項目名を変更する必要がある場合があります。

3 データの変換

3 データの変換
アプリでデータを使用する前に、データロードエディターでデータを変換および操作できます。

データ操作のメリットの1 つは、ファイルからデータのサブセットのみをロードするという選択ができることです。つま

り、テーブルからいくつかの列だけを選択してより効率的にデータを処理できます。また、データを複数回に渡り

ロードして、生データをいくつかの新しい論理テーブルに分化することもできます。さらに、複数のソースからデータ

をロードして、Qlik Sense にある1 つのテーブルに結合することも可能です。

次の演習では、Crosstable プレフィックスを使用してデータをロードする方法について説明します。また、テーブ

ルの結合、レコード間関数 (Peek およびPrevious) の使用、While Load を使って同じ行を何回もロードする

方法についても学習します。

3.1 Crosstable プレフィックスの使用

クロス集計は、ヘッダーデータが直交する2 つのリストに値のマトリックスを持つ一般的なテーブル

の種類で、データのクロス集計を使用する場合は、常にCrosstable プレフィックスを使用して、

データの変換や希望する項目の作成ができます。

Crosstable プレフィックス

次のProduct テーブルには、Month (月) ごとに1 列、Product (製品) ごとに1 行あります。

製品 Jan 2014 Feb 2014 Mar 2014 Apr 2014 May 2014 Jun 2014

A 100 98 100 83 103 82

B 284 279 297 305 294 292

C 50 53 50 54 49 51

Product テーブル

このテーブルをロードすると、Product とそれぞれの月に1 項目ずつ存在するテーブルが生成されます。

Product 項目と、月ごとに1 項目ずつあるProduct テーブル

3 データの変換

このデータを分析する場合は、すべての数値が 1 つの項目にあり、別の項目にすべての月がある方がはるかに

容易です。この例の場合、各カテゴリ(Product, Month, Sales) ごとに1 つの列を持つ3 列構成のテーブルで

す。

Product、Month、Sales の各項目があるProduct テーブル

Crosstable プレフィックスは、データを1 列がMonth、もう1 つがSales を表すテーブルに変換します。要する

に、項目名を取得して項目値に変換します。

次の手順を実行します。

1. 新しいアプリを作成し、高度なスクリプトチュートリアルという名前を付けます。

2. データロードエディターで新しいスクリプトセクションを追加します。

3. Product セクションに名前を付けます。

4. 右のメニューの [AttachedFiles] で、[データを選択] をクリックします。

5. Product.xlsx をアップロードして選択します。

6. [Select data from] (データの選択元) ウィンドウで、Product テーブルを選択します。

[項目名] で [埋め込まれた項目名] が選択されていることを確認します。このオプションが選

択されていると、データをロードする際にテーブルの項目名が含まれます。

7. [スクリプトを挿入] をクリックします。

これで、スクリプトは次のようになります。

LOAD

Product,

"Jan 2014",

"Feb 2014",

"Mar 2014",

"Apr 2014",

"May 2014",

"Jun 2014"

FROM [lib://AttachedFiles/Product.xlsx]

(ooxml, embedded labels, table is Product);

8. [データのロード] をクリックします。

9. [データモデルビューア] を開きます。データモデルは次のようになります。

3 データの変換

Product 項目と、月ごとに1 項目ずつあるProduct テーブル

10. データロードエディターで、[Product] タブをクリックします。

11. LOAD ステートメントの上に次の内容を入力します。

CrossTable(Month, Sales)

12. [データのロード] をクリックします。

13. [データモデルビューア] を開きます。データモデルは次のようになります。

Product、Month、Sales の各項目があるProduct テーブル

通常入力データには、内部キー (上の例ではProduct) として機能する修飾子項目の列が 1 つだけあり

ます。ですが、ここでは複数持つことが可能です。その場合は、LOAD ステートメントの属性項目の前に

すべての補助項目をリストし、Crosstable プレフィックスの3 番目のパラメータを補助項目の数を定義

するために使用する必要があります。Crosstable キーワードの前に先行するLOAD またはプレフィックス

を持つことはできません。ただし、自動連結を使用できます。

Qlik Sense 内のテーブルで、データは次のようになります。

3 データの変換

Crosstable プレフィックスを使用してロードしたデータのテーブル

これで、たとえば、データを使用して棒グラフを作成できるようになりました。

Crosstable プレフィックスを使用してロードしたデータの棒グラフ

Crosstable について詳しくは、Qlik Community の次のブログ投稿を参照してください。

「Crosstable ロード」。動作はQlikView の状況に応じて説明されていますが、ロジックはQlik
Sense にもそのまま当てはまります。

3 データの変換

属性項目では、数値解釈は行われません。これは、列のヘッダーが月である場合、自動的に解釈されないこと

を意味します。この回避策としては、Crosstable プレフィックスを使用して一時的なテーブルを作成し、次の例

のように2 回目のパススルーを実行して解釈を行います。

これはあくまで例です。Qlik Sense で行う付随する演習はありません。

tmpData:

Crosstable (MonthText, Sales)

LOAD Product, [Jan 2014], [Feb 2014], [Mar 2014], [Apr 2014], [May 2014], [Jun 2014]

FROM ...

Final:

LOAD Product,

Date(Date#(MonthText,'MMM YYYY'),'MMM YYYY') as Month,

Sales

Resident tmpData;

Drop Table tmpData;

メモリキャッシュのクリア

作成するテーブルを削除し、メモリキャッシュをクリアできます。前のセクションのようにデータを一時的なテーブル

にロードした場合、不要になったら削除しなければなりません。例 :

DROP TABLE Table1, Table2, Table3, Table4;

DROP TABLES Table1, Table2, Table3, Table4;

項目を削除することもできます。例 :

DROP FIELD Field1, Field2, Field3, Field4;

DROP FIELDS Field1, Field2, Field3, Field4;

DROP FIELD Field1 from Table1;

DROP FIELDS Field1 from Table1;

ご覧のとおり、キーワードTABLE とFIELD は単数形でも複数形でも使用できます。

3.2 Join とKeep を使用したテーブルの結合
結合とは2 つのテーブルを1 つにまとめる操作です。結合により生成されるテーブルのレコードは元のテーブルの

レコードを組み合わせたもので、通常、生成されたテーブル内の特定の組み合わせに関係する2 つのレコードが

1 つ以上の共通項目に共通の値を持っています。そのため、これを「自然結合」と呼びます。Qlik Sense では、

結合はスクリプトで実行され、論理テーブルが作成されます。

スクリプト内で事前にテーブルを結合することができます。その場合、Qlik Sense ロジックは、個々のテーブルで

はなく、結合の結果である単一の内部テーブルを処理するようになります。この処理が必要な場合もあります

が、次のような欠点があります。

l ロードされたテーブルのサイズが大きくなり、Qlik Sense の動作が遅くなることがあります。

l 元のテーブル内のレコード数を取得できなくなり、情報の一部が失われる可能性があります。

Keep 機能は、Qlik Sense にテーブルを保存する前に、2 つのテーブルの片方または両方をテーブルデータの共

通部分に縮小させる効果を持ち、明示的な参加を使用する必要がある場合の数を減らすために設計されて

います。

3 データの変換

このマニュアルでは、結合という用語は、内部テーブルが作成される前に行われる結合という意味で

使用されています。しかし、内部テーブルが作成された後で行われる関連付けも、基本的には結

合です。

Join
結合を作成する最も簡単な方法は、スクリプトでJoin プレフィックスを使用することです。これは、内部テーブル

を別の名前のテーブルまたは最後に作成されたテーブルに結合します。この結合は外部結合になり、2 つのテー

ブルからのすべての可能な値の組み合わせを作成します。

LOAD a, b, c from table1.csv;

join LOAD a, d from table2.csv;

その結果の内部テーブルには、項目 a、b、c、d が含まれています。レコード数は、2 つのテーブルの項目値に

よって異なります。

結合される項目の名前は、完全に同じである必要があります。結合される項目数は任意です。通

常、テーブルには 1 つまたは複数の共通する項目があります。共通する項目が 1 つもない場合、

テーブルのデカルト積は生成されません。すべての項目が共通することも考えられますが、通常これ

は意味がありません。Join ステートメントで、先にロードされたテーブルの名前が指定されていない

限り、Join プレフィックスは最後に作成されたテーブルを使用します。そのため、2 つのステートメント

の順序を任意に指定することはできません。

Join の使用

Qlik Sense スクリプト言語における明示的な Join プレフィックスは、2 つのテーブルの完全な結合を実行しま

す。その結果、1 つのテーブルが生成されます。このような結合を行うと、テーブルが非常に大きくなることがよくあ

ります。

次の手順を実行します。

1. 高度なスクリプトチュートリアルアプリを開きます。

2. データロードエディターで新しいスクリプトセクションを追加します。

3. このセクションをTransactions と呼びます。

4. 右のメニューの [AttachedFiles] で、[データを選択] をクリックします。

5. Transactions.csv をアップロードして選択します。

[項目名] で [埋め込まれた項目名] が選択されていることを確認します。このオプションが選

択されていると、データをロードする際にテーブルの項目名が含まれます。

6. [Select data from] (データの選択元) ウィンドウで、[スクリプトを挿入] をクリックします。

7. Salesman.xlsx をアップロードして選択します。

3 データの変換

8. [Select data from] (データの選択元) ウィンドウで、[スクリプトを挿入] をクリックします。

これで、スクリプトは次のようになります。

LOAD

"Transaction ID",

"Salesman ID",

Product,

"Serial No",

"ID Customer",

"List Price",

"Gross Margin"

FROM [lib://AttachedFiles/Transactions.csv]

(txt, codepage is 28591, embedded labels, delimiter is ',', msq);

LOAD

"Salesman ID",

Salesman,

"Distributor ID"

FROM [lib://AttachedFiles/Salesman.xlsx]

(ooxml, embedded labels, table is Salesman);

9. [データのロード] をクリックします。

10. [データモデルビューア] を開きます。データモデルは次のようになります。

データモデル:Transactions、Salesman、およびProduct テーブル

ただし、Transactions とSalesman のテーブルを分離すると、必要な結果が得られないことがあります。この2
つのテーブルは結合した方がよいでしょう。

3 データの変換

次の手順を実行します。

1. 結合したテーブルの名前を設定するには、最初のLOAD ステートメントの上に次の行を追加します。

Transactions:

2. Transactions とSalesman テーブルを結合するには、2 番目のLOAD ステートメントの上に次の行を

追加します。

Join(Transactions)

これで、スクリプトは次のようになります。

Transactions:

LOAD

"Transaction ID",

"Salesman ID",

Product,

"Serial No",

"ID Customer",

"List Price",

"Gross Margin"

FROM [lib://AttachedFiles/Transactions.csv]

(txt, codepage is 28591, embedded labels, delimiter is ',', msq);

Join(Transactions)

LOAD

"Salesman ID",

Salesman,

"Distributor ID"

FROM [lib://AttachedFiles/Salesman.xlsx]

(ooxml, embedded labels, table is Salesman);

3. [データのロード] をクリックします。

4. [データモデルビューア] を開きます。データモデルは次のようになります。

3 データの変換

データモデル: Transactions およびProduct テーブル

これでTransactions およびSalesman テーブルのすべての項目が、単一のTransactions テーブルに

結合されました。

Join をいつ使用するかについて詳しくは、Qlik Community の次のブログ投稿を参照してください。

「To Join or not to join」 (結合するかしないか)、「Mapping as an ALternative to Joining」 (結
合の代わりにマッピングする)。動作はQlikView の状況に応じて説明されていますが、ロジックは

Qlik Sense にもそのまま当てはまります。

Keep
Qlik Sense の主な機能の1 つは、テーブルを結合する代わりにテーブル間に関連付けを作成できることです。こ

れにより、メモリの使用量が削減され、速度が向上し、柔軟性が大幅に高まります。Keep の機能は、明示的

な結合の使用回数を減らす目的で設計されています。

2 つのLOAD またはSELECT ステートメント間のKeep プレフィックスは、Qlik Sense にテーブルが保存される前

に、2 つのテーブルの片方または両方をテーブルデータの共通部分に縮小させます。Keep プレフィックスの前に

は、Inner、Left、またはRight のいずれかのキーワードを置く必要があります。テーブルからのレコードの選択は、

対応する結合と同じ方法で行われます。ただし、2 つのテーブルは結合されず、個別のテーブルとしてそれぞれ

異なる名前でQlik Sense に保存されます。

Inner
データロードスクリプトのJoin およびKeep プレフィックスの前に、プレフィックス Inner を置くことができます。

Join の前に使用すると、2 つのテーブル間の結合が内部結合であると指定されます。結果のテーブルには、両

方のテーブルのすべてのデータセットのうち、2 つのテーブル間で共通するものだけが含まれます。

Keep の前に使用すると、Qlik Sense に保存される前に、2 つのテーブルは互いの共通部分に縮小されます。

これらの例では、Table1 とTable2 のソーステーブルを使用しています。

これらはあくまで例です。Qlik Sense で行う付随する演習はありません。

A B

1 aa

2 cc

3 ee

Table 1

A C

1 xx

4 yy

Table2

Inner Join
まず、これらのテーブルで Inner Join を実行すると、VTable が作成されます。このテーブルには2 つのテーブルの

データを結合した結果、1 行だけ、両方のテーブルに存在するレコードだけが表示されます。

VTable:

SELECT * from Table1;

inner join SELECT * from Table2;

A B C

1 aa xx

VTable

Inner Keep
それに対して、Inner Keep を実行すると、2 つのテーブルが残ります。これらの2 つのテーブルは、共通項目 A
を介して関連付けられます。

VTab1:

SELECT * from Table1;

VTab2:

inner keep SELECT * from Table2;

A B

1 aa

VTab1

A C

1 xx

VTab2

Left
データロードスクリプトのJoin およびKeep プレフィックスの前に、プレフィックス left を置くことができます。

Join の前に使用すると、2 つのテーブル間の結合が左結合であると指定されます。結果のテーブルには、1 つ目

のテーブルのすべてのデータセットと2 つのテーブルの共通するものだけが含まれます。

Keep の前に使用すると、Qlik Sense に保存される前に、2 つ目のテーブルが 1 つ目のテーブルとの共通部分に

縮小されます。

これらの例では、Table1 とTable2 のソーステーブルを使用しています。

A B

1 aa

2 cc

3 ee

Table1

A C

1 xx

4 yy

Table2

まず、これらのテーブルでLeft Join を実行すると、VTable が生成されます。このテーブルには、Table1 のすべて

の行に加え、Table2 の一致行の項目が含まれています。

VTable:

SELECT * from Table1;

left join SELECT * from Table2;

A B C

1 aa xx

2 cc -

3 ee -

VTable

それに対して、Left Keep を実行すると、2 つのテーブルが残ります。これらの2 つのテーブルは、共通項目 A を

介して関連付けられます。

VTab1:

SELECT * from Table1;

VTab2:

left keep SELECT * from Table2;

A B

1 aa

2 cc

3 ee

VTab1

A C

1 xx

VTab2

Right
Qlik Sense スクリプト言語内のJoin およびKeep プレフィックスの前に、プレフィックスright を置くことができま

す。

Join の前に使用すると、2 つのテーブル間の結合が右結合であると指定されます。結果のテーブルには、2 つ目

のテーブルのすべてのデータと2 つのテーブルで共通するものだけが含まれます。

Keep の前に使用すると、Qlik Sense に保存される前に、1 つ目のテーブルが2 つ目のテーブルとの共通部分に

縮小されます。

これらの例では、Table1 とTable2 のソーステーブルを使用しています。

A B

1 aa

2 cc

3 ee

Table1

A C

1 xx

4 yy

Table2

まず、これらのテーブルでRight Join を実行すると、VTable が生成されます。このテーブルには、Table2 のすべ

ての行に加え、Table1 の一致行の項目が含まれています。

VTable:

SELECT * from Table1;

right join SELECT * from Table2;

A B C

1 aa xx

4 - yy

VTable

それに対して、Right Keep を実行すると、2 つのテーブルが残ります。これらの2 つのテーブルは、共通項目 A
を介して関連付けられます。

VTab1:

SELECT * from Table1;

VTab2:

right keep SELECT * from Table2;

A B

1 aa

VTab1

A C

1 xx

4 yy

VTab2

3.3 レコード間関数の使用 :Peek、Previous、およびExists
これらの関数は、現在のレコードの評価に以前にロードされたデータのレコード値が必要な場合に

使用します。

チュートリアルのここでは、Peek()、Previous()、およびExists() 関数を検証します。

Peek()
Peek()は、すでにロードされている行に対してテーブルの項目値を返します。行番号は、テーブルと同様に指

定できます。行番号が指定されていない場合は、最後にロードされたレコードが使用されます。

構文 :
Peek(fieldname [ , row [ , tablename ] ] )

row は整数にする必要があります。0 は最初のレコード、1 は2 番目のレコードを示し、以下同様に表されま

す。負の数は、テーブルの最後から見た順序を表します。-1 は、読み取られた最後のレコードを示します。

row が指定されていない場合は、-1 として処理されます。

Tablename は、末尾にコロンが付いていない、テーブルのラベルです。tablename が指定されていない場合

は、現在のテーブルとして処理されます。LOADステートメント以外で使用する、または他のテーブルを参照する

場合は、tablename が含まれている必要があります。

Previous()
Previous()は、where節のために破棄されなかった以前の入力レコードのデータを使用して、expr数式の値を

算出します。内部テーブルの最初のレコードの場合は、NULL を返します。

構文 :
Previous(expression)

Previous() 関数をネストすることで、さらに前のレコードにアクセスすることができます。データは入力ソースから

直接取得されるため、Qlik Sense にまだロードされていない (つまり、関連するデータベースに保存されていなく

ても) 項目を参照することもできます。

Exists()
Exists()は、特定の項目値がデータロードスクリプトの項目にすでにロードされているかどうかを決定します。こ

の関数はTRUE または FALSE を返すため、LOADステートメントまたは IFステートメントのwhere句で使用

できます。

構文 :
Exists(field [, expression ] )

項目は、スクリプトによってこれまでにロードされたデータの中になければなりません。Expression は、指定した項

目において、検索対象項目値として評価される数式です。省略されると、指定された項目の現在のレコード値

が使用されます。

Peek() およびPrevious() の使用

もっとも簡単な形式では、Peek() およびPrevious() はテーブル内で特定の値を指定するために使用されま

す。この演習でロードするEmployees テーブル内のサンプルデータを示します。

Date Hired Terminated

1/1/2011 6 0

2/1/2011 4 2

3/1/2011 6 1

4/1/2011 5 2

従業員テーブルのサンプルデータ

現在、これは月、採用、退職などのデータのみを集計しており、Peek() とPrevious() 関数を使ってこれに

Employee Count とEmployee Var の項目を追加して、従業員総数の月ごとの変化を確認します。

次の手順を実行します。

1. 高度なスクリプトチュートリアルアプリを開きます。

2. データロードエディターで新しいスクリプトセクションを追加します。

3. このセクションをEmployees と呼びます。

4. 右のメニューの [AttachedFiles] で、[データを選択] をクリックします。

5. Employees.xlsx をアップロードして選択します。

[Field names] で [Embedded field names] が選択されていることを確認します。このオプ

ションが選択されていると、データをロードする際にテーブルの項目名が含まれます。

6. [Select data from] (データの選択元) ウィンドウで、[スクリプトを挿入] をクリックします。

これで、スクリプトは次のようになります。

LOAD

"Date",

Hired,

Terminated

FROM [lib://AttachedFiles/Employees.xlsx]

(ooxml, embedded labels, table is Sheet1);

7. スクリプトを、次のように修正します。
[Employees Init]:

LOAD

rowno() as Row,

Date(Date) as Date,

Hired,

Terminated,

If(rowno()=1, Hired-Terminated, peek([Employee Count], -1)+(Hired-Terminated)) as

[Employee Count]

FROM [lib://AttachedFiles/Employees.xlsx]

(ooxml, embedded labels, table is Sheet1);

Excel シートのDate 項目の日付は、MM/DD/YYYY の書式です。日付がこの書式を使用してシステム

変数から正しく解釈されていることを確認するには、Date 関数をDate 項目に適用します。

チュートリアル- スクリプトでの次のステップ- Qlik Sense, May 2024 22



3 データの変換

Peek() 関数が、定義済みの項目にロードされた値を特定します。この数式では、最初にrowno() が 1
と等しいかどうかを確認します。1 と等しい場合、Employee Count は存在しないため、Hired -
Terminated の差をその項目に入力します。

rowno() が 1 より大きい場合は、前月のEmployee Count を確認し、その数値をその月のHired -
Terminated の値に加算します。

Peek() 関数に(-1) を使用していることに注意してください。これは、Qlik Sense に現在のレコードを上

回るレコードを確認するよう指示しています。(-1) が指定されていない場合は、Qlik Sense は前のレコー

ドを確認することを意味します。

8. 次の内容をスクリプトの最後に追加します。

[Employee Count]:

LOAD

Row,

Date,

Hired,

Terminated,

[Employee Count],

If(rowno()=1,0,[Employee Count]-Previous([Employee Count])) as [Employee Var]

Resident [Employees Init] Order By Row asc;

Drop Table [Employees Init];

Previous() 関数が、定義済みの項目にロードされた最新の値を特定します。この数式では、最初に

rowno() が 1 と等しいかどうかを確認します。1 と等しい場合は、前月のEmployee Count のレコードが

存在せず、それゆえEmployee Var は存在しないため、値として単に0 を入力します。

rowno()が 1 より大きい場合、Employee Varが存在していることを示すので、先月のEmployee
Count を確認し、その数値を現在の月のEmployee Count から差し引いてEmployee Var 項目の値

を作成します。

これで、スクリプトは次のようになります。

[Employees Init]:

LOAD

rowno() as Row,

Date(Date) as Date,

Hired,

Terminated,

If(rowno()=1, Hired-Terminated, peek([Employee Count], -1)+(Hired-Terminated)) as

[Employee Count]

FROM [lib://AttachedFiles/Employees.xlsx]

(ooxml, embedded labels, table is Sheet1);

[Employee Count]:

LOAD

Row,

Date,

Hired,

Terminated,

[Employee Count],

If(rowno()=1,0,[Employee Count]-Previous([Employee Count])) as [Employee Var]

Resident [Employees Init] Order By Row asc;

Drop Table [Employees Init];

9. [データのロード] をクリックします。

アプリ概要の新しいシートで、テーブルの列としてDate, Hired, Terminated, Employee Count および

Employee Var を使用してテーブルを作成します。この結果、テーブルは次のようになります。

スクリプトでPeek とPrevious を使用した後のテーブル

Peek() とPrevious() を使用すると、テーブル内の定義済み行を対象にすることができます。この2 つの関数の

最大の違いは、Peek() 関数ではユーザーがスクリプトにあらかじめロードされていない項目を参照することができ

ますがPrevious() 関数はロード済みの項目しか参照できない点です。Previous() は、LOAD ステートメントへ

の入力で値を返し、Peek() は LOAD ステートメントの出力で値を返します。(RecNo() とRowNo() の違いと同

じ。) これは、Where 節を使用すると、2 つの関数が異なる動作をすることを意味します。

Previous() 関数は、ユーザーが現在の値と前の値を表示する必要がある場合に、より適しています。例では、

月によって異なる従業員の分散状況を計算しています。

Peek() 関数は、ユーザーがテーブルに事前にロードされていない項目を対象とする場合、または特定の行を対

象とする必要がある場合のいずれにも、より適しています。これは、前月のEmployee Count をピークすることで

Employee Count を計算し、今月の雇用者と退職者数の差異を加算している例で示しています。Employee
Count は、オリジナルファイルの項目ではないことに注意が必要です。

Peek() とPrevious() を使用するタイミングについて詳しくは、Qlik Community の次のブログ投稿

を参照してください。「Peek() vs Previous() – When to Use Each」 動作はQlikView の状況に

応じて説明されていますが、ロジックはQlik Sense にもそのまま当てはまります。

Exists() の使用

Exists() 関数は、スクリプトにおいてWhere 節とともに使用され、データモデルにすでにロードされている関連

データがある場合にデータをロードします。

次の例では、Dual() 関数を使用して文字列に数値を割り当てています。

次の手順を実行します。

1. 新しいアプリを作成し、名前を付けます。

2. データロードエディターで新しいスクリプトセクションを追加します。

3. このセクションをPeople と呼びます。

4. 次のスクリプトを入力します。

//Add dummy people data

PeopleTemp:

LOAD * INLINE [

PersonID, Person

1, Jane

2, Joe

3, Shawn

4, Sue

5, Frank

6, Mike

7, Gloria

8, Mary

9, Steven,

10, Bill

];

//Add dummy age data

AgeTemp:

LOAD * INLINE [

PersonID, Age

1, 23

2, 45

3, 43

4, 30

5, 40

6, 32

7, 45

8, 54

9,

10, 61

11, 21

12, 39

];

//LOAD new table with people

People:

NoConcatenate LOAD

PersonID,

Person

Resident PeopleTemp;

Drop Table PeopleTemp;

//Add age and age bucket fields to the People table

Left Join (People)

LOAD

PersonID,

Age,

If(IsNull(Age) or Age='', Dual('No age', 5),

If(Age<25, Dual('Under 25', 1),

If(Age>=25 and Age <35, Dual('25-34', 2),

If(Age>=35 and Age<50, Dual('35-49' , 3),

If(Age>=50, Dual('50 or over', 4)

))))) as AgeBucket

Resident AgeTemp

Where Exists(PersonID);

DROP Table AgeTemp;

5. [データのロード] をクリックします。

スクリプトでは、AgeおよびAgeBucket項目がロードされるのは、PersonID がすでにデータモデルにロー

ドされている場合に限られます。

AgeTemp テーブルには年齢のリスト(PersonID 11 と12) がありますが、これらの ID はデータモデル

(Peopleテーブル) にはロードされておらず、Where Exists(PersonID) 節によって除外されていることに注

意してください。この節はまた、のように書くこともできます。Where Exists(PersonID, PersonID) をクリッ

クします。

スクリプトの出力は次のようになります。

スクリプトでExists を使用した後のテーブル

PersonID がAgeTemp データモデルにロード済みのテーブルにない場合、Age とAgeBucket項目は

People テーブルには結合されません。Exists()関数を使用すると、データモデル内に孤立したレコード/
データが残りません。つまり、AgeとAgeBucket項目に関連する従業員はいないことになります。

6. 新しいシートを作成し、名前を付けます。

7. その新しいシートを開き、[シートの編集] をクリックします。

8. 軸 AgeBucket を含むシートに標準的なテーブルを追加し、ビジュアライゼーションに「Age Groups」とい
う名前を付けます。

9. 軸 AgeBucket とメジャーCount([AgeBucket]) を持つシートに棒グラフを追加します。ビジュアライゼー

ションにNumber of people in each age group という名前を付けます。

10. テーブルと棒グラフのプロパティを希望に応じて調整し、[完了] をクリックします。

シートは次のようになります。

年齢別にグループ化したシート

Dual() 関数は、文字列に数値を割り当てる必要があるスクリプトやチャートの数式で役に立ちます。

上記のスクリプトでは、年齢をロードしているアプリケーションがあり、この年齢をバケットに入れ、この年齡バケット

と実年齡に基づくビジュアライゼーションを作成します。従業員 25 歳未満、25～35 歳といったバケットがありま

す。Dual() 関数を使用すると、後でリストボックスやチャート内で年齡バケットをソートするために便利な数値

を、年齡バケットに割り当てることができます。つまりアプリシートでは、ソートによってリストの終端に「No age
(年齡なし)」が追加されます。

Exists() とDual() について詳しくは、Qlik Community の次のブログ投稿を参照してください。

「Dual & Exists – Useful Functions」 (Dual とExists – 便利な関数)

3.4 間隔の一致と反復ロード

Intervalmatch プレフィックスをLOAD またはSELECT ステートメントで使用すると、不連続の数

値が 1 つ以上の数値間隔にリンクされます。これはとても有用な機能で、実稼働環境などで使

用できます。

IntervalMatch() プレフィックスの使用

もっとも基本的な間隔一致は、1 つのテーブルに数値や日付 (イベント) のリストが存在し、2 つ目のテーブルに

間隔のリストがある場合です。目的は、この2 つのテーブルをリンクさせることです。一般的に、これは多対多の

関係です。つまり、間隔には多数の日付が属し、日付は多数の間隔に属する場合があります。これを解決する

には、2 つのオリジナルテーブル間の橋渡しとなるブリッジテーブルを作成する必要があります。これには、さまざ

まな方法があります。

Qlik Sense におけるこの問題を解決するもっとも簡単は方法は、IntervalMatch() プレフィックスをLOAD また

はSELECT ステートメントの前に配置する方法です。LOAD/SELECT ステートメントには、間隔を定義する

From とTo という2 つの項目のみを含める必要があります。IntervalMatch() プレフィックスは、プレフィックスのパ

ラメータで指定された、ロード済みの間隔と前にロードされた数値項目間のすべての組み合わせを生成します。

次の手順を実行します。

1. 新しいアプリを作成し、名前を付けます。

2. データロードエディターで新しいスクリプトセクションを追加します。

3. このセクションをEvents と呼びます。

4. 右のメニューの [AttachedFiles] で、[データを選択] をクリックします。

5. Events.txt をアップロードして選択します。

6. [Select data from] (データの選択元) ウィンドウで、[スクリプトを挿入] をクリックします。

7. Intervals.txt をアップロードして選択します。

8. [Select data from] (データの選択元) ウィンドウで、[スクリプトを挿入] をクリックします。

9. スクリプトで、1 つ目のテーブルにEvents、2 つ目のテーブルに Intervals という名前を付けます。

10. スクリプトの最後に IntervalMatch を追加して、最初の2 つのテーブルをリンクさせる3 つ目のテーブルを

作成します。

BridgeTable:

IntervalMatch (EventDate)

LOAD distinct IntervalBegin, IntervalEnd

Resident Intervals;

11. これで、スクリプトは次のようになります。

Events:

LOAD

EventID,

EventDate,

EventAttribute

FROM [lib://AttachedFiles/Events.txt]

(txt, utf8, embedded labels, delimiter is '\t', msq);

Intervals:

LOAD

IntervalID,

IntervalAttribute,

IntervalBegin,

IntervalEnd

FROM [lib://AttachedFiles/Intervals.txt]

(txt, utf8, embedded labels, delimiter is '\t', msq);

BridgeTable:

IntervalMatch (EventDate)

LOAD distinct IntervalBegin, IntervalEnd

Resident Intervals;

12. [データのロード] をクリックします。

13. [データモデルビューア] を開きます。データモデルは次のようになります。

データモデル: Events、BridgeTable、Intervals、および$Syn1 テーブル

データモデルには複合キー (IntervalBegin とIntervalEnd 項目) が含まれ、これはQlik Sense 合成

キーとして現れます。

基本的なテーブルは、

l イベントごとに正確に1 つのレコードを含む、Events テーブル。

l 間隔ごとに正確に1 つのレコードを含む、Intervals テーブル。

l イベントと間隔の組み合わせごとに正確に1 つのレコードを含み、上記 2 つのテーブルをリンクさせ

ているブリッジテーブル。

間隔の重複によって、イベントに複数の間隔が付属している場合があります。そして1 つの間隔に複数

のイベントが付属している場合もあります。

このデータモデルは、データの正規化やコンパクト化という意味においては最適です。Events テーブルと

Intervals テーブルは、ともに不変で元の数と同じレコード数が含まれています。Count(EventID) など、

これらのテーブルで行われるQlik Sense のすべての計算は、正しく動作し、評価されます。

IntervalMatch() について詳しくは、Qlik Community の次のブログ投稿を参照してください。

IntervalMatch() の使用

While ループの使用とIterNo() の反復ロード

間隔の下限と上限間の可算値を生成するWhile ループとIterNo() を使用すると、ブリッジテーブルと同様の結

果が得られます。

LOAD ステートメント内のループは、次のようにWhile 節を使って作成できます。例 :

LOAD Date, IterNo() as Iteration From … While IterNo() <= 4;

このような LOAD ステートメントは、各入力レコードを何度もループし、While 節の数式が true である限りロード

は何度も続きます。IterNo() 関数は、最初の反復では「1」、2 回目の反復では「2」というように値を返します。

間隔には IntervalID というプライマリキーがあるため、スクリプトにおける唯一の違いはブリッジテーブルを作成す

る方法です。

次の手順を実行します。

1. 既存のBridgetable ステートメントを、次のスクリプトと置き換えます。

BridgeTable:

LOAD distinct * Where Exists(EventDate);

LOAD IntervalBegin + IterNo() - 1 as EventDate, IntervalID

Resident Intervals

While IntervalBegin + IterNo() - 1 <= IntervalEnd;

2. [データのロード] をクリックします。

3. [データモデルビューア] を開きます。データモデルは次のようになります。

データモデル: Events、BridgeTable、および Intervals テーブル

一般に、3 つのテーブルを有するソリューションが最善です。これは間隔とイベント間に多対多の関係が

許容されるからです。ですが多くの場合、1 つのイベントには単一の間隔が付属します。この場合、ブリッ

ジテーブルは必要ではありません。IntervalID は、イベントテーブルに直接保存されます。これを実現す

るにはいくつかの方法がありますが、もっとも有用な方法はBridgetable とEvents テーブルを結合するこ

とです。

4. 次のスクリプトを、スクリプトの最後に追加します。

Join (Events)

LOAD EventDate, IntervalID

Resident BridgeTable;

Drop Table BridgeTable;

5. [データのロード] をクリックします。

6. [データモデルビューア] を開きます。データモデルは次のようになります。

データモデル: Events および Intervals テーブル

開区間と閉区間

間隔が開いているまたは閉じているかどうかは、エンドポイントが間隔に含まれているかどうかで判断します。

l エンドポイントが含まれている場合、それは閉区間です。

[a,b] = {x ∈ ℝ ∣ a ≤ x ≤ b}

l エンドポイントが含まれていない場合、それは開区間です。

]a,b[ = {x ∈ ℝ ∣ a < x < b}

l 1 つのエンドポイントが含まれている場合、それは半開区間です。

[a,b[ = {x ∈ ℝ ∣ a ≤ x < b}

間隔が重複している場合や数値が2 つ以上の間隔に付属している場合、通常は閉区間を使用する必要が

あります。

しかし、間隔を重複させずに、1 つの間隔にのみ数値を付属させたい場合もあります。よって、1 つのポイントが 1
つの間隔の終端にあり、同時に次の間隔の冒頭にある場合は問題になります。この値を持つ数値は、両方の

間隔に属することになります。つまり、このような場合は半開区間を使用します。

この問題の現実的な解決策は、すべての間隔の終点地から少量を差し引き、閉じているが重複しない間隔を

作成することです。数値が日付だった場合、日付の最後のミリ秒を返すDayEnd() 関数を使用するのが、もっ

とも簡単な方法です。

Intervals:

LOAD…, DayEnd(IntervalEnd – 1) as IntervalEnd From Intervals;

手動で少量を差し引くこともできます。これを実施すると、値が2 進 52 桁の有効数字 (10 進 14 桁) に四捨

五入されるため、差し引いた量が小さすぎないことを確認してください。小さすぎる値を使用すると、差異に意味

がなくなり、元の数値が使用されます。

4 データクレンジング
Qlik Sense にロードするソースデータが、Qlik Sense アプリで希望どおりにならない場合があります。Qlik
Sense は、データを希望する形式に変換するための関数とステートメントのホストを提供しています。

Qlik Sense スクリプトを実行する際にマッピングを使用すると、項目値や名前の置換または変更ができます。つ

まり、マッピングによりデータをより整合性のあるデータに整理したり、項目値の一部または全部と置換することが

可能になります。

データを異なるテーブルからロードする際、同じことを表す項目値に、一貫した名前が付いているとは限りませ

ん。この一貫性の欠如は関連性を見落とす原因にもなるため、対処が必要です。この問題は、項目値を比較

するマッピングテーブルを作成することで解決できます。

4.1 マッピングテーブル
Mapping load またはMapping select からロードされたテーブルは、他のテーブルとは異なる方法で処理されま

す。これらのテーブルはメモリの別の領域に保存され、スクリプトの実行時にマッピングテーブルとしてのみ使用さ

れます。スクリプトが実行されると、これらのテーブルは自動的に削除されます。

ルール:
l マッピングテーブルは、1 列目に比較値、2 列目にマッピング値を含む2 列構成でなければなりません。

l 2 つの列に名前を付ける必要がありますが、名前と列に含まれる値に関連性はありません。これらの列

の名前は、通常の内部テーブルの項目名とは一切関係がありません。

4.2 Mapping 関数とステートメント
このチュートリアルでは、次のマッピング関数/ステートメントについて説明します。

•Mapping プレフィックス

• ApplyMap()

• MapSubstring()

•Map … Using ステートメント

•Unmap ステートメント

4.3 マッピングプレフィックス
Mapping プレフィックスは、マッピングテーブルを作成するためにスクリプトで使用します。マッピングテーブルは、

ApplyMap() 関数またはMapSubstring() 関数、Map … Using ステートメントと共に使用できます。

次の手順を実行します。

1. 新しいアプリを作成し、名前を付けます。

2. データロードエディターで新しいスクリプトセクションを追加します。

3. このセクションをCountries と呼びます。

4. 次のスクリプトを入力します。

CountryMap:

MAPPING LOAD * INLINE [

Country, NewCountry

U.S.A., US

U.S., US

United States, US

United States of America, US

];

CountryMap テーブルには、2 つの列(Country およびNewCountry) が保存されます。Country 列に

は、様々な表記方法でCountry 項目に入力された国が保存されます。NewCountry 列には、値を

マッピングする方法に関する情報が保存されます。このマッピングテーブルは、一貫性のあるUS という国

の値をCountry 項目に保存するために使用されます。つまり、U.S.A. とCountry 項目に保存されて

も、それはUS としてマッピングされます。

4.4 ApplyMap() 関数
ApplyMap() を使用して、前に作成したマッピングテーブルに基づいて項目のデータを置き換えます。マッピング

テーブルは、ApplyMap() 関数を使用する前にロードする必要があります。ロードするData.xlsx テーブルのデー

タは次のようになります。

データテーブル

ID 名前 国 コード

1 John Black USA SDFGBS1DI

2 Steve Johnson U.S. 2ABC

3 Mary White 米国 DJY3DFE34

4 Susan McDaniels u DEF5556

5 Dean Smith US KSD111DKFJ1

国が様々な表記方法で入力されているのが分かります。国項目に一貫性を持たせるために、マッピングテーブ

ルをロードし、ApplyMap()関数を使用します。

次の手順を実行します。

1. 上で作成したスクリプトの下で、Data.xlsx を選択してロードし、スクリプトを挿入します。

2. 新しく作成したLOAD ステートメントの上に以下を入力します。

Data:

これで、スクリプトは次のようになります。

CountryMap:

MAPPING LOAD * INLINE [

Country, NewCountry

U.S.A., US

U.S., US

United States, US

United States of America, US

];

Data:

LOAD

ID,

Name,

Country,

Code

FROM [lib://AttachedFiles/Data.xlsx]

(ooxml, embedded labels, table is Sheet1);

3. Country,を含む行を次のように修正します。

ApplyMap('CountryMap', Country) as Country,

ApplyMap() 関数の最初のパラメータには、単一引用符で囲まれたマップ名があります。2 番目のパラ

メータは、置換されるデータを含む項目です。

4. [データのロード] をクリックします。

この結果、テーブルは次のようになります。

ApplyMap() 関数を使用してロードしたデータのテーブル

様々なつづりで表記されたUnited States が、すべてUS に変換されます。1 つのレコードはつづりが間

違っているため、ApplyMap() 関数はこれを項目値に変更していません。ApplyMap() 関数を使用する

際に、3 番目のパラメータを使ってマッピングテーブルに一致する値がない場合のデフォルトの数式を追

加することができます。

5. 'US'をApplyMap() 関数の3 番目のパラメータとして追加し、国名が正しく入力されなかった場合に対

応します。

ApplyMap('CountryMap', Country, 'US') as Country,

これで、スクリプトは次のようになります。

CountryMap:

MAPPING LOAD * INLINE [

Country, NewCountry

U.S.A., US

U.S., US

United States, US

United States of America, US

];

Data:

LOAD

ID,

Name,

ApplyMap('CountryMap', Country, 'US') as Country,

Code

FROM [lib://AttachedFiles/Data.xlsx]

(ooxml, embedded labels, table is Sheet1);

6. [データのロード] をクリックします。

この結果、テーブルは次のようになります。

ApplyMap 関数を使用してロードしたデータのテーブル

ApplyMap() について詳しくは、Qlik Community の次のブログ投稿を参照してください。「Don't
join - use Applymap instead」 (結合しない- Applymap で代用)

4.5 MapSubstring() 関数
MapSubstring() 関数は、項目の部分的なマッピングを可能にします。

ApplyMap() で作成したテーブルに数値を文字列として記入したい場合、MapSubstring() 関数を使って数値

を文字列に置換できます。

これを実行するには、まずマッピングテーブルを作成する必要があります。

次の手順を実行します。

1. CountryMap セクションの末尾、かつDataセクションの前に次のスクリプト行を追加します。

CodeMap:

MAPPING LOAD * INLINE [

F1, F2

1, one

2, two

3, three

4, four

5, five

11, eleven

];

CodeMapテーブルでは、数値の1～5 と11 がマッピングされます。

2. スクリプトのData セクションで、Codeステートメントを次のように修正します。

MapSubString('CodeMap', Code) as Code

これで、スクリプトは次のようになります。

CountryMap:

MAPPING LOAD * INLINE [

Country, NewCountry

U.S.A., US

U.S., US

United States, US

United States of America, US

];

CodeMap:

MAPPING LOAD * INLINE [

F1, F2

1, one

2, two

3, three

4, four

5, five

11, eleven

];

Data:

LOAD

ID,

Name,

ApplyMap('CountryMap', Country, 'US') as Country,

MapSubString('CodeMap', Code) as Code

FROM [lib://AttachedFiles/Data.xlsx]

(ooxml, embedded labels, table is Sheet1);

3. [データのロード] をクリックします。

この結果、テーブルは次のようになります。

MapSubString 関数を使用してロードしたデータのテーブル

数字がCode 項目で文字列に置換されています。ID=3 および ID=4 のように数字が 1 回以上現れる

場合は、テキストもまた繰り返されます。ID=4.Susan McDaniels は、そのコードに6 がありました。6 は

CodeMap テーブルにマッピングされていないので、変更されません。ID=5、Dean Smith のコードには

111 がありました。これは、'elevenone' としてマッピングされています。

MapSubstring() について詳しくは、Qlik Community の次のブログ投稿を参照してください。

「Mapping … and not the geographical kind」 (地理学ではないマッピング)

4.6 Map … Using
Map … Using ステートメントもまた、項目へのマッピングに使用できます。ただし、ApplyMap() とは機能が少し

異なります。ApplyMap() は項目名が発生するたびにマッピングを実行しますが、Map … Using は、値が内部

テーブルの項目名に保存された場合にマッピングを行います。

例を挙げて見てみましょう。スクリプトでCountry 項目を複数回ロードしており、項目がロードされる度にマッピン

グを実行したい場合を想定します。このチュートリアルですでに説明したようにApplyMap() 関数が使用可能で

あり、Map … Using を使用することもできます。

Map … Using が使用されると、項目が内部テーブルに保存される際にマップが項目に適用されます。つまり、

以下の例のようにData1 テーブルのCountry 項目に対するマッピングが実行されますが、Data2 テーブルの

Country2 項目に対するマッピングは実行されません。これは、Map … Using ステートメントがCountry という

名前の項目にのみ適用されることが理由です。Country2 項目が内部テーブルに保存されると、Country とい

う名前ではなくなります。Country2 テーブルに対するマッピングを実行したい場合は、ApplyMap() 関数を使用

する必要があります。

Unmap ステートメントはMap … Using ステートメントを終了させます。つまりCountry がUnmap ステートメン

トの後にロードされると、CountryMap は適用されません。

次の手順を実行します。

1. Data テーブルのスクリプトを次のスクリプトで置き換えます。

Map Country Using CountryMap;

Data1:

LOAD

ID,

Name,

Country

FROM [lib://AttachedFiles/Data.xlsx]

(ooxml, embedded labels, table is Sheet1);

Data2:

LOAD

ID,

Country as Country2

FROM [lib://AttachedFiles/Data.xlsx]

(ooxml, embedded labels, table is Sheet1);

UNMAP;

2. [データのロード] をクリックします。

この結果、テーブルは次のようになります。

Map … Using 関数を使用してロードしたデータのテーブル

5 階層データの処理
階層構造は、すべてのビジネスインテリジェンスソリューションの重要な部分で、異なる粒度レベ

ルを含む軸を説明するために使用します。シンプルで直感的なものもあれば、正しくモデルを構築

するためによく考える必要がある複雑なものもあります。

メンバーは階層の最上部から最下部に向かって、段階的に詳細になっていきます。例えば、軸に市場、国、

州、市というレベルがある場合、南北アメリカ大陸は階層の最上部に位置し、米国は2 番目のレベル、カリフォ

ルニアは3 番目、サンフランシスコは最下部のレベルに表示されます。カリフォルニアは米国よりも範囲が狭く、サ

ンフランシスコはカリフォルニアよりさらに詳細度が増しています。

リレーショナルモデルにおける階層の保存は、複数のソリューションがある場合、よく問題になります。これには、

いくつかのアプローチがあります。

l 横方向の階層

l 隣接リストモデル

l 経路列挙方法

l 入れ子集合モデル

l 先祖リスト

このチュートリアルの目的においては、クエリーで直接使用できる形式の階層を提示するために、先祖リストを作

成します。その他のアプローチの詳細については、Qlik Community を参照してください。

5.1 Hierarchy プレフィックス
Hierarchy プレフィックスはスクリプトコマンドで、隣接するノードテーブルをロードするLOAD またはSELECT ス

テートメントの前に置きます。LOAD ステートメントには、少なくとも3 つの項目(ノードの一意のキーであるID、

親ノードへの参照、名前) が必要です。

プレフィックスはロードされたテーブルを展開ノードテーブルに変換します。テーブルには多数の追加列 (階層の

各レベルごとに1 つ) が含まれています。

次の手順を実行します。

1. 新しいアプリを作成し、名前を付けます。

2. データロードエディターで新しいスクリプトセクションを追加します。

3. このセクションをWine と呼びます。

4. 右のメニューの [AttachedFiles] で、[データを選択] をクリックします。

5. Winedistricts.txt をアップロードして選択します。

6. [Select data from] (データの選択元) ウィンドウで、Lbound とRBound 項目を選択解除してロードさ

れないようにします。

7. [スクリプトを挿入] をクリックします。

8. LOAD ステートメントの上に次の内容を入力します。

Hierarchy (NodeID, ParentID, NodeName)

これで、スクリプトは次のようになります。

Hierarchy (NodeID, ParentID, NodeName)

LOAD

NodeID,

ParentID,

NodeName

FROM [lib://AttachedFiles/Winedistricts.txt]

(txt, utf8, embedded labels, delimiter is '\t', msq);

9. [データのロード] をクリックします。

10. 結果テーブルを確認するには、データモデルビューアの [プレビュー] セクションを使用してください。

結果の展開ノードテーブルには、ソーステーブルとまったく同じ数のレコード(ノードごとに1 つ) がありま

す。展開ノードテーブルは、リレーショナルモデルの階層を解析する、次のような多数の要件を満たして

いるため非常に実用的です。

l すべてのノード名は 1 つの同じ列に存在するため、検索に使用できます。

l さらに、さまざまなノードレベルがそれぞれ 1 つの項目 (ドリルダウングループで、またはピボット

テーブルの軸として使用可能な項目) に展開されます。

l さらに、異なるノードレベルがそれぞれ 1 つの項目に展開されるため、項目をドリルダウングループ

で使用することができます。

l ノード固有のパスを含むように作成して、すべての祖先を正しい順番でリストすることができます。

l ノードの深度 (ルートからの距離) も含めるよう作成することができます。

この結果、テーブルは次のようになります。

Hierarchy プレフィックスを使用してロードしたサンプルデータのテーブル

5.2 HierarchyBelongsTo プレフィックス
Hierarchy プレフィックス同様、HierarchyBelongsTo プレフィックスはスクリプトコマンドで、隣接するノードテー

ブルをロードするLOAD またはSELECT ステートメントの前に置きます。

ここでもまた、LOAD ステートメントには少なくとも3 つの項目(ノードの一意のキーであるID、親ノードへの参

照、名前) が必要です。プレフィックスはロードされたテーブルを展開ノードテーブルに変換します。テーブルには

あらゆる先祖の組み合わせおよび、個別レコードとして子孫がリストされています。よって、特定ノードの先祖ま

たは子孫はすべて簡単に検索できます。

次の手順を実行します。

1. データロードエディターで、Hierarchy ステートメントを次のように修正します。

HierarchyBelongsTo (NodeID, ParentID, NodeName, BelongsToID, BelongsTo)

2. [データのロード] をクリックします。

3. 結果テーブルを確認するには、データモデルビューアの [プレビュー] セクションを使用してください。

先祖テーブルは、リレーショナルモデルの階層を解析する、次のような多数の要件を満たしています。

l ノードID が単一ノードを表す場合、先祖 ID 階層のツリー全体およびサブツリーを表します。

l すべてのノード名は、ノードとしての役割およびツリーとして役割の両方に存在し、両方とも検索

に使用できます。

l ノード深度とサブツリーのルートからの距離である先祖深度の間の深度差異を含むよう、作成す

ることもできます。

この結果、テーブルは次のようになります。

HierarchyBelongsTo プレフィックスを使用してロードしたデータのテーブル

承認

階層を承認に使用することは、珍しいことではありません。1 つの例としては、組織的な階層が挙げられます。各

マネージャーが、すべてのサブ部門含め、自分の部門に関連するあらゆる権限を持っている必要があります。で

すが、必ずしも他の部門を管理する権限を持つべきではない、というわけではありません。

組織的な階層の例

これは、閲覧を許可される組織内のサブツリーは人によって異なるということを意味します。認証テーブルは、例

えば、以下のようになります。

ACCESS NTNAME PERSON POSITION PERMISSIONS

ユーザー ACME\JRL John CPO HR

ユーザー ACME\CAH Carol CEO CEO

ユーザー ACME\JER James エンジニアリング部長 エンジニアリング

ユーザー ACME\DBK Diana CFO 財務

ユーザー ACME\RNL Bob COO 売上

ユーザー ACME\LFD Larry CTO 製品

承認テーブル

この場合、Carol はCEO とその下の関係するすべての組織を閲覧でき、Larry はProduct 組織を、James は

Engineering 組織のみを閲覧できます。

多くの場合、階層は隣接ノードテーブルに保存されます。この例では、これを解決するために、

HierarchyBelongsTo を使用して隣接ノードテーブルをロードし、先祖項目Tree に名前を付けます。

Section Access を使用する場合は、Tree の大文字コピーをロードし、この新しい項目をPERMISSIONS とし

ます。最後に、承認テーブルをロードする必要があります。最後の2 つのステップは、次のスクリプト行を使用して

実行することが可能です。TempTrees テーブルはHierarchyBelongsTo ステートメントによって作成される

テーブルです。

これはあくまで例です。Qlik Sense で行う付随する演習はありません。

Trees:

LOAD *,

Upper(Tree) as PERMISSIONS

Resident TempTrees;

Drop Table TempTrees;

Section Access;

Authorization:

LOAD ACCESS,

NTNAME,

UPPER(Permissions) as PERMISSIONS

From Organization;

Section Application;

この例では、次のデータモデルが生成されます。

データモデル:Authorization、Trees、Fact、およびNodes テーブル

6 QVD ファイル
QVD (QlikView Data) ファイルは、Qlik Sense またはQlikView からエクスポートされたデータのテーブルを含む

ファイルです。QVD はネイティブのQlik 形式で、Qlik Sense またはQlikView でしか書き込みと読み出しを行え

ません。ファイル形式は、Qlik Sense スクリプトからデータを読み取る際の速度について最適化されていますが、

それでも非常にコンパクトです。QVD ファイルからのデータの読み取りは、他のデータソースから読み取る場合よ

りも、一般に10～100 倍速くなっています。

QVD ファイルは、標準 (高速) と最適化 (超高速) の2 つのモードで読み取ることができます。使用されるモード

は、Qlik Sense スクリプトエンジンによって自動的に決定されます。最適化モードは、ロードされたすべての項目

が変換 (項目に対して実行される式) なしに読み取られる場合にのみ使用できます。ただし、項目名は変更す

ることができます。Where 節は、Qlik Sense にレコードを解凍させ、最適化されたロードを無効にします。

QVD ファイルは、厳密に1 つのデータテーブルを保持し、次の3 つの部分で構成されます。

l テーブル内の項目や、後続情報およびその他のメタデータのレイアウトを記述するXML ヘッダー (UTF-8
文字セット)。

l バイト埋め込み形式のシンボルテーブル。

l ビット埋め込み形式の実際のテーブルデータ。

QVD ファイルの用途はさまざまですが、主に4 つの目的で使用されます。複数の目的を一度に達成できる場

合もあります。

l データロード速度の向上

QVD ファイルの入力データの変更されない部分または変化が遅い部分をバッファリングすることで、大き

なデータセットに対するスクリプトの実行が大幅に高速化します。

l データベースサーバーの負荷の減少

外部データソースから取得するデータ量を大幅に削減できます。これにより、外部データベースおよび

ネットワークトラフィックの負荷が減少します。さらに、複数のQlik Sense スクリプトが同じデータを共有

する場合は、ソースデータベースからQVD ファイルに一度データをロードするだけで済みます。他のアプリ

ケーションは、このQVD ファイルから同じデータを使用できます。

l 複数のQlik Sense アプリケーションからのデータ統合

Binary スクリプトステートメントでは、単一のQlik Sense アプリケーションからのみ別のアプリケーションに

データをロードできます。しかし、QVD ファイルを使用すれば、Qlik Sense スクリプトは、任意の数のQlik
Sense アプリケーションからデータを統合できます。これにより、さまざまな部署の類似データを統合する

など、アプリケーションの可能性が広がります。

l 増分ロード

多くの場合、QVD の機能を活用することで、増大するデータベースから新しいレコードだけをロードする

増分ロードを簡単に実行できます。

Qlik コミュニティがQlik アプリケーションの自動化 を使用してQVD のロード時間を改善する方法に

ついては、「≤ 自動化を使用してQVD を分割し、リロードを改善する方法」を参照してくださ

い。

6.1 QVD ファイルの作成
QVD ファイルは2 つの方法で作成できます。

l Qlik Sense スクリプトでStore コマンドを使用した明示的な作成と命名。

スクリプトで、以前に読み取ったテーブルまたはその一部を、選択した場所にある明示的に指定された

ファイルにエクスポートするように記述します。

l スクリプトからの自動作成とメンテナンス。

load または select ステートメントの前にBuffer プレフィックスを付けると、Qlik Sense は自動的にQVD
ファイルを作成します。このQVD ファイルは一定の条件下で、データのリロード時に元のデータソースの

代わりに使用できます。

読み込み時間など、生成されたQVD ファイルに違いはありません。

Store

このスクリプトステートメントは、明示的に命名されたQVD、CSV、txt ファイルを作成します。

構文 :
Store[ *fieldlist from] table into filename [ format-spec ];

項目のエクスポートは 1 つのデータテーブルからのみ行えます。複数のテーブルの項目がエクスポートされる場

合、エクスポートされるデータテーブルを作成するためには、事前にスクリプトで明示的に結合されている必要が

あります。

テキストの値はUTF-8 形式でCSV ファイルに出力されます。区切り文字を指定できます (LOADを参照)。
CSV ファイルへのstore ステートメントは、BIFF エクスポートをサポートしていません。

例 :

Store mytable into [lib://AttachedFiles/xyz.qvd];

Store * from mytable into [lib://FolderConnection/xyz.qvd];

Store myfield from mytable into 'lib://FolderConnection/xyz.qvd';

Store myfield as renamedfield, myfield2 as renamedfield2 from mytable into

[lib://AttachedFiles/xyz.qvd];

Store mytable into 'lib://FolderConnection/myfile.txt';

Store * from mytable into 'lib://FolderConnection/myfile.csv';

次の手順を実行します。

1. 高度なスクリプトチュートリアルアプリを開きます。

2. Product スクリプトセクションをクリックします。

3. 次の内容をスクリプトの最後に追加します。

Store * from Product into [lib://AttachedFiles/ProductData.qvd](qvd);

これで、スクリプトは次のようになります。

CrossTable(Month, Sales)

LOAD

Product,

"Jan 2014",

"Feb 2014",

"Mar 2014",

"Apr 2014",

"May 2014"

FROM [lib://AttachedFiles/Product.xlsx]

(ooxml, embedded labels, table is Product);

Store * from Product into [lib://AttachedFiles/ProductData.qvd](qvd);

4. [データのロード] をクリックします。

これでProduct.qvd ファイルがファイルリストに表示されるようになります。

Crosstableスクリプトの結果であるこのデータファイルは、各カテゴリ(Product, Month, Sales) ごとに1
つの列を持つ3 列構成のテーブルです。このデータファイルは、Product スクリプトセクション全体を置換

するために使用できます。

6.2 QVD ファイルからのデータのロード
Qlik Sense は次の方法でQVD ファイルの読み込みやアクセスを実行します。

l 明示的なデータソースとしてのQVD ファイルのロード。QVD ファイルは、その他のテキストファイル (csv、
fix、dif、biff など) と同様に、Qlik Sense スクリプトの load ステートメントで参照できます。

例 :

LOAD * from 'lib://FolderConnection/xyz.qvd' (qvd);

LOAD fieldname1, fieldname2 from [lib://FolderConnection/xyz.qvd] (qvd);

LOAD fieldname1 as newfieldname1, fieldname2 as newfieldname2 from

[lib://AttachedFiles/xyz.qvd](qvd);

l バッファ済みのQVD ファイルの自動ロード。load または select ステートメントでbuffer プレフィックスを使

用する場合、読み取り用の明示的なステートメントは必要ありません。Qlik Sense は元のLOAD ス

テートメントまたはSELECT ステートメントを使用してデータを取得するのではなく、QVD ファイルからどの

範囲のデータを使用するかを決定します。

l スクリプトからのQVD ファイルへのアクセス。(QVD で始まる) いくつかのスクリプト関数を使用して、QVD
ファイルのXML ヘッダーにあるデータのさまざまな情報を取得できます。

次の手順を実行します。

1. Product スクリプトセクションのスクリプト全体をコメントアウトします。

2. 次のスクリプトを入力します。

Load * from [lib://AttachedFiles/ProductData.qvd](qvd);

3. [データのロード] をクリックします。

データはQVD ファイルからロードされます。

データロード進捗状況ウィンドウ

増分ロードでのQVD ファイルの使用について詳しくは、Qlik Community の次のブログ投稿を参照

してください。「Overview of Qlik Incremental Loading」 (Qlik 増分ロードの概要)

Buffer
QVD ファイルは、Buffer プレフィックスを使用して、自動的に作成、管理することができます。このプレフィックス

は、スクリプトのLOAD ステートメントおよびSELECT ステートメントのほとんどで使用できます。つまり、ステート

メントの結果をキャッシュ/バッファする際には、QVD ファイルが使用されます。

構文 :
Buffer [ (option [ , option])] ( loadstatement | selectstatement )

option::= incremental | stale [after] amount [(days | hours)]

オプションを使用していない場合、最初のスクリプト実行で作成されたQVD バッファが無限に使用されます。

Buffer load * from MyTable;

stale [after] amount [(days | hours)]

Amount は期間を指定する数字で、Decimals を使用できます。単位が省略されている場合は、日数と見な

されます。

通常、stale after オプションは、元のデータに一般的なタイムスタンプがないデータベースソースで使用します。

stale after 節は、QVD バッファが作成されてから有効期限切れになるまでの期間を指定します。それまでの

間、QVD バッファがデータソースとして使用され、期間終了後は元のデータソースが使用されます。その後、

QVD バッファファイルが自動更新され、新しい期間が開始します。

Buffer (stale after 7 days) load * from MyTable;

Incremental

incremental オプションを使用すると、基底ファイルの一部のみを読み取る機能が有効になります。以前のファイ

ルサイズは、QVD ファイルのXML ヘッダーに保存されます。これは、ログファイルで特に便利です。過去にロード

されたレコードは、すべてQVD ファイルから読み取られますが、以降の新しいレコードについては元のソースから

読み取った上でQVD ファイルを更新します。

incremental オプションは LOAD ステートメントとテキストファイルでのみ使用でき、古いデータが変更されたり削

除されている場合、増分ロードは使用できませんのでご注意ください。

Buffer (incremental) load * from MyLog.log;

通常、バッファを作成したアプリのスクリプトで一切参照されなくなったり、アプリが存在しなくなると、QVD バッファ

は削除されます。Store ステートメントは、バッファのコンテンツをQVD またはCSV ファイルとして保持したい場合

に使用します。

次の手順を実行します。

1. 新しいアプリを作成し、名前を付けます。

2. データロードエディターで新しいスクリプトセクションを追加します。

3. 右のメニューの [AttachedFiles] で、[データを選択] をクリックします。

4. Cutlery.xlsx をアップロードして選択します。

5. [Select data from] (データの選択元) ウィンドウで、[スクリプトを挿入] をクリックします。

6. Load ステートメントの項目をコメントアウトして、Load ステートメントを次のように変更します。

Buffer LOAD *

これで、スクリプトは次のようになります。

Buffer LOAD *

// "date",

// item,

// quantity

FROM [lib://AttachedFiles/Cutlery.xlsx]

(ooxml, embedded labels, table is Sheet1);

7. [データのロード] をクリックします。

初めてデータをロードするときは、Cutlery.xlsx からロードされます。

データロード進捗状況ウィンドウ

Buffer ステートメントはQVD ファイルを作成し、このファイルをQlik Sense に保存します。Qlik Sense
Enterprise on Windows 展開では、Qlik Sense サーバー上のディレクトリに保存されます。

8. [データのロード] を再度クリックします。

9. 今回は、初めてデータをロードしたときにBuffer ステートメントによって作成されたQVD ファイルからデー

タがロードされます。

データロード進捗状況ウィンドウ

6.3 お疲れ様でした!
以上でチュートリアルは終了です。ここでご紹介した基本情報をQlik Sense でのスクリプト作成にお役立てくだ

さい。ウェブサイトでは、ご利用可能なトレーニングについての情報を公開しています。ぜひご覧ください。

